The Volume Allowance Formula as a Guide to Non-Haemic Solution Administration

Michael S. Vinas, RRT, CCPT, CCP

Vinas and Associates
Nashville, TN

Keywords: Non-haemic solution, extracorporeal circuit, target hematocrit, volume allowance

Abstract
We have developed a formula sequence which calculates the predicted blood volume, hemodilutional hematocrit and conjectively figures the amount of non-haemic solution margin allowable before reaching a minimally acceptable or 'target' hematocrit in the adult patient.

Another application of the formula calculates the value for the amount of prime reduction to prevent the addition of blood to the ECC priming volume. The formula also provides the figures for the amount of blood to add to the prime of the extracorporeal circuit in order to maintain a target hematocrit.

Introduction
Many institutions have escalating concerns toward the administration of blood products. Complications such as hepatitis, AIDS, antibody-antigen reactions, etc. could be contained with a conservative blood product administration program and would mean savings in medical supply and administrative costs.

The practice of blood conservation, especially in the field of extracorporeal circulatory technology, necessitates the need for various computations, which include predicted blood volume and approximate hematocrit adjusted for hemodilution with known amounts of non-haemic solutions. We have enhanced the application of these formulas to include a volume allowance calculation which provides us with valuable information as it pertains to the conduct of perfusion, the priming volume of the extracorporeal circuit, non-haemic solution and/or blood product administration requirements, if indicated.

Example 1
Step 1: Calculate the predicted blood volume.
\[ 70 \text{ kg} \times 70 \text{ ml/kg} \]
\[ 4900 \text{ ml.-predicted blood volume (PBV)} \]

Step 2: Compute the patients red cell mass (RCM1).
\[ 4900 \text{ ml.-PBV} \times \frac{38}{100} \]
\[ 1862 \text{ ml. RCM1} \]

Step 3: Figure the hemodilutional hematocrit (HCT2).
\[ .28 = \frac{1862 (RCM1)}{6700 (PBV+Prime= \text{Total System Volume})} \]
\[ .28 \times 100 = 28\% (HCT2) \]
Step 4: Calculate the RCM at target or minimally accepted hemodilutional hematocrit value.

$$6700 \text{ TSV (Total System Volume)} \times 0.21$$

$$= 1407 = \text{RCM2}$$

Step 5: Figure the RCM difference and divide by the target hematocrit expressed as a decimal.

$$1862 - \text{RCM1}$$

$$- 1407 - \text{RCM2}$$

$$455 - \text{RCM difference}$$

$$455 = 2167 \text{ mL volume allowance}$$

$$.21 \text{ (target HCT) (HCT2)}$$

The various steps in the volume allowance calculation may be condensed and expressed in the following equation:

$$\frac{[(\text{KG} \times 70) \times (\text{HCT1}/100)] - [\text{TSV} \times (\text{HCT2}/100)]}{(\text{HCT2}/100)}$$

or

$$\frac{(\text{RCM1} - \text{RCM2})}{\text{HCT2}}$$

Our ongoing hemodilutional research tentatively reveals that predicted blood volumes, compared to a computer regression model, may vary on average of ±20% after standard deviation, especially for the average adult female population; therefore, the volume allowance value should be discounted by administering half the calculated value, then reevaluating the hemodilutional status with an updated hematocrit.

The volume allowance formula is especially appreciable in the smaller, more anemic patient. For example, a female patient weighing 50 kg with a pre-bypass hematocrit of 32%, an extracorporeal circuit volume of 1500 ml and a minimum acceptable hematocrit of 21% would present the following:

Example 2

$$1120 \text{ RCM1}$$

$$- 1050 \text{ RCM2}$$

$$70 \text{ RCM difference}$$

$$70 \frac{\text{delta RCM}}{= 333 \text{ ml volume allowance}}$$

$$.21 \text{ (Fractional HCT2)}$$

In Example 2, the patient should be entitled to approximately 333 ml of non-haemic solution(s) before reaching the target hematocrit or minimal accepted hematocrit. Half this amount should be administered and a reevaluation undertaken. Also, the relatively small volume allowance should alert the clinician that blood products may need to be administered and should be immediately accessible.

The formula may be adjusted to provide a value for blood administration. For example, if the patient had a hematocrit of 28% and required a prime volume of 1800 ml, the RCM1 would calculate to be 980 ml and the adjusted hematocrit figure to be 18.5%.

Example 3

$$980 \text{ RCM1}$$

$$- 1113 \text{ RCM2}$$

$$- 133 \text{ delta RCM}$$

$$- 133 = 633 \text{ ml prime reduction}$$

$$.21 \text{ HCT2}$$

Analytically, if the volume allowance formula figures a negative number, the formula becomes a volume reduction instrument indicating the amount of prime to reduce to obtain the desired hematocrit without the addition of blood.

If the amount of prime can not be altered, then the absolute value of delta RCM is taken for the amount of RBC's to administer.

It has been our experience that an average unit of packed red blood cells (PRBC) preserved with ADSOL contains a volume of approximately 300 ml with a hematocrit approaching 50%. CPD-A units differ by having an approximate hematocrit of 70% but are usually reconstitituted before administration. The RCM difference divided by 150 should provide the needed number of ADSOL units versus 210 for non-reconstituted CPD-A units of PRBCs. Therefore:

$$133 = 0.9 \text{ ADSOL units-PRBC}$$

$$150$$

$$133 = 0.6 \text{ CPD-A units-PRBC}$$

$$210$$

Calculations such as those presented in this article may be tedious with a hand-held calculator and subject to operator error; therefore, it is strongly recommended to incorporate a computer for assistance.

We have provided a token computer program written in BASIC computer language applicable with an IBM compatible computer. The program is also compatible with an EPSON HX-20 portable computer. The program incorporates error trapping to prevent input data that is not compatible to routine situations.

```
10 REM VOLUME ALLOWANCE FORMULA
20 REM WRITTEN BY MICHAEL S VINAS
30 REM VINAS and ASSOCIATES
40 REM NASHVILLE TN 37215
50 ' 60 REM DATA INPUT SECTION
70 CLS
80 GOSUB 790
90 INPUT "WEIGHT (KGS.) " ;KG
```
100 IF KG< 35 OR KG> 120 THEN 70
110 ' 
120 CLS
130 GOSUB 790
140 INPUT "HCT (%)"; HCT1
150 IF HCT1< 25 OR HCT1>50 THEN 120
160 ' 
170 CLS
180 GOSUB 790
190 INPUT "PRIME (ML.)"; PRIME
200 IF PRIME<1200 OR PRIME >3000 THEN 170
210 ' 
220 CLS
230 GOSUB 790
240 INPUT "DESIRED HCT (%)"; HCT2
250 IF HCT1< HCT2 THEN 120
260 IF HCT2<21 OR HCT2>38 THEN 220
270 CLS
280 ' 
290 REM CALCULATION SECTION
300 BLDVOL=KG*70
310 RCM1=BLDVOL*(HCT1/100)
320 TSV=BLDVOL+PRIME
330 ADJHCT=(RCM1/TSV)*100
340 ' 
350 ' 
360 RCM2=TSV*(HCT2/100)
370 RCM3=RCM1-RCM2
380 VOLALLOW=RCM3/(HCT2/100)
390 ADJHCT=INT ((ADJHCT*10)+.5)/10
400 VOLALLOW=INT (VOLALLOW)
410 REDUCE=ABS (VOLALLOW)
420 ' 
430 PRBC=RCM2-RCM1
440 ADSOLUNIT=PRBC/150
450 ADSOLUNIT=INT ((ADSOLUNIT*10)+.5)/10
460 CPDAUNIT=PRBC/120
470 CPDAUNIT=INT ((CPDAUNIT*10)+.5)/10
480 ' 
490 REM DATA OUTPUT SECTION
500 ' 
510 CLS
520 GOSUB 790
530 PRINT "DATE= " ; DATES
540 PRINT "TIME= " ; TIMES
550 PRINT
560 PRINT "BLOOD VOLUME (ML)= " ; BLDVOL
570 PRINT "ADJUSTED HCT. (%)= " ; ADJHCT
580 PRINT "VOL. ALLOW. (ML)= " ; VOLALLOW
590 ' 
600 IF VOLALLOW>0 THEN 690
610 ' 
620 PRINT "REDUCE PRIME (ML) " ; REDUCE
630 PRINT
640 PRINT " <OR>
650 PRINT
660 PRINT "PRBC NEEDED= " ; PRBC
670 PRINT "ADSOL UNIT(S)= " ; ADSOLUNIT
680 PRINT "CPD-A UNIT(S)= " ; CPDAUNIT
690 ' 
700 FOR I=1 TO 3
710 PRINT ""
720 NEXT I
730 ' 
740 PRINT ; TAB (28) '<DEPRESS ANY KEY TO RESUME>
750 A$=INKEY$: IF A$=" " THEN 750 ELSE 50
760 ' 
770 END
780 ' 
790 ' 
800 CLS
810 PRINT ; TAB (29) "VOLUME ALLOWANCE PROGRAM"
820 FOR I=1 TO 4
830 PRINT
840 NEXT I
850 RETURN
860 END

Abbreviations

ADJHCT = Adjusted hematocrit
ADSOL = ADSOL units of packed red blood cells
BLDVOL = Patient blood volume
CPDA = CPD-A units of packed red blood cells
HCT1 = Patient hematocrit
HCT2 = Desired hematocrit
KG = Kilogram
PRBC = Packed red blood cells needed
PRIME = Priming volume of the extracorporeal circuit
REduce = The amount of prime to reduce sans blood
RCM1 = Red cell mass of the patient
RCM2 = Red cell mass of the total system
RCM3 = delta RCM1-RCM2
TSV = Total system volume (BLDVOL+PRIME)
VOLALLOW = Volume allowance of non-haemic solution(s)

Conclusion

Formulas that calculate patient blood volume and adjusted
hemodilutional hematocrits provide a guide to non-haemic
volume allowance as well as minimum blood administration.
These are recommended utilities for those institutions that are
actively engaged in the conservation of blood administration as
a reduction to possible complications associated with the use of
blood and blood byproducts, as well as institutional cost
containment. The volume allowance equation is an
enhancement to these formulas and provides values concerning
allowance of non-haemic solution or ECC prime reduction.

References

1. Berger, Edward C. The Physiology of Adequate Perfusion. The C.V.
2. Reed C and Stafford,T. Cardiopulmonary Bypass-2nd Ed., Texas
   Medical Press, Houston, TX, 1985. p. 131.
3. Miripol JE. ADSOL preservation solution-A new additive for the
   extended storage of red blood cells. Fenwall Laboratories, Deerfield,
   IL, 1986.